
EDIT VARIABLE 
 
TAA productivy tools 
 
CMD : EDTVAR               
 100 /*------------------------------------------------------------------------------------*/  
 200 /* EDIT VARIABLE - Cmd EDTVAR                                                         */ 
 300 /*------------------------------------------------------------------------------------*/  
 400 /* PARMS PGM(EDTVAR) ALLOW((*IPGM)(*BPGM)) PRDLIB(ZSYS)                               */ 
 600 /* La Cmd EDTVAR transfert une variable décimale dans une variable de 22 caractères   */ 
 700 /* avec codes édition et symbole courant en option                                    */ 
1000 /* Le CPP est EDTVAR                                                                 */ 
1200 /*-----------------------------------------------------------------------------------*/ 
1300         CMD PROMPT('Editer une variable numérique') 
1400         PARM KWD(CHROUT) TYPE(*CHAR) LEN(22) RTNVAL(*YES) +  
1500              MIN(1) PROMPT('Variable pour Retour (22)')  
1700         PARM KWD(NUMINP) TYPE(*DEC) LEN(15 0) MIN(1) +  
1800              PROMPT('Nombre à formater (15,0)') 24/11/92 
1900         PARM KWD(NBRDEC) TYPE(*DEC) LEN(1 0) DFT(0) +  
2000              RANGE(0 9) PROMPT('Nombre de décimales (1,0)')  
2200         PARM KWD(EDTCDE) TYPE(*CHAR) LEN(1) RSTD(*YES) +  
2300              DFT(J) VALUES(J K L M W Y Z) PROMPT('Code Edition')  
2500         PARM KWD(LDGSYM) TYPE(*CHAR) LEN(4) RSTD(*YES) +  
2600              DFT(*NO) VALUES('*YES' '*NO') PROMPT('Symbole monétaire')2700  
 
 
CL : EDTVAR               
             PGM        PARM(&CHARVAR &NUMVAR &NBRDEC &EDTCDE &LDGSYM)           
                                                                                 
 DCLPARM:    DCL        &CHARVAR *CHAR 22             /* Variable en Retour */   
             DCL        &NUMVAR  *DEC (15 0)          /* Nombre en entrée   */   
             DCL        &NBRDEC  *DEC  (1 0)          /* Nombre de décimales */  
             DCL        &EDTCDE  *CHAR  1             /* Code édition */         
             DCL        &LDGSYM  *CHAR  4             /* Symbole à traiter  */   
 DCLWORK:    DCL        &CURSYM  *CHAR  1             /* Symbole */              
             DCL        &NUMVAR2 *DEC (15 0)          /* Copie de &NUMVAR */     
                                                                                 
             DCL        &DECCHAR *CHAR  1 (',')       /* Car. décimal  */        
             DCL        &SEPCHAR *CHAR  1 ('.')       /* Car. Milliers */        
             DCL        &NUMVARC *CHAR 15             /* Entrée en Caractères */ 
             DCL        &PRTSW   *CHAR  1             /* Switch Imprimé */       
             DCL        &X       *DEC  (3 0) (1)      /* Index Entrée */         
             DCL        &Y       *DEC  (3 0) (1)      /* Index Sortie */         
             DCL        &Z       *DEC  (3 0)          /* Index Séparateur */     
             DCL        &DECFMT  *CHAR  1             /* Format décimal */       
                                                                                 
             IF         ((&EDTCDE *EQ 'W') *OR (&EDTCDE *EQ 'Y') +               
                          *OR (&EDTCDE *EQ 'Z')) DO             /* W Y Z edit */ 
                                                                                 
               IF       (&NBRDEC *NE 0)   SNDPGMMSG +                            
                            MSGID(CPF9898) MSGF(QCPFMSG) MSGDTA('Si +            
                            EDTCDE est W, Y ou Z, le Nombre de +                 
                            décimales doit être à 0') MSGTYPE(*ESCAPE)           
             ENDDO                                              /* W Y Z edit */ 
                                                                                 
             CHGVAR     &CHARVAR ' '    /* Initialisation variable de Retour */  
                                                                                 
         /* Si Code édition K, M ou Z, Test valeur à 0 */                        
             IF         ((&EDTCDE *EQ 'K') *OR (&EDTCDE *EQ 'M') +               
                          *OR (&EDTCDE *EQ 'Z')) DO                              
                IF         (&NUMVAR *EQ 0) GOTO ENDPGM                           
             ENDDO                                                               
                                                                                 
         /* Rtv format décimal et init valeurs */                                
             RTVSYSVAL  SYSVAL(QDECFMT) RTNVAR(&DECFMT)                          
                                                                                 
             IF  (&DECFMT *EQ '1') DO  /* Format USA */                          
                CHGVAR     &DECCHAR '.'                                          
                CHGVAR     &SEPCHAR ','                                          
             ENDDO                                                               

         /* Initialisation zones travail et index */                             
             CHGVAR     &Z (1 + &NBRDEC)                 /* Index séparation */  
             CHGVAR     &NUMVAR2 &NUMVAR                 /* Variable travail */  
             IF         (&NUMVAR *LT 0)  +                                       
             CHGVAR     &NUMVAR2  (0 - &NUMVAR)          /* Inversion signe */   
             CHGVAR     &NUMVARC &NUMVAR2              /* Changement en *Char */ 
                                                                                 
         /* Mise en place du Symbole monétaire si nécessaire */                  
             IF         (&LDGSYM *EQ '*YES') DO            /* Currency symbol */ 
                RTVSYSVAL  QCURSYM RTNVAR(&CURSYM)                               
                CHGVAR     %SST(&CHARVAR &Y 1) &CURSYM           /* Insertion */ 
                CHGVAR     &Y (&Y + 1)                    /* Màj index sortie */ 
             ENDDO                                                               
                                                                                 
         /* Mise en place de chaque caractère */                                 
LOOP:        IF         ((16 - &X) *EQ &NBRDEC) DO     /* Emplacement décimale*/ 
               CHGVAR     %SST(&CHARVAR &Y 1) &DECCHAR      /* Insertion déc. */ 
               CHGVAR     &Y (&Y + 1)                     /* Màj index sortie */ 
               CHGVAR     &PRTSW 'Y'                                             
             ENDDO                                                               
                                                                                 
             IF         (&DECFMT *EQ 'J') DO              /* Format J - ldg 0 */ 
                IF         ((15 - &X) *EQ &NBRDEC) CHGVAR &PRTSW 'Y'             
             ENDDO                                                               
                                                                                 
         /* Si EDTCDE W ou Y Switch Prt après le 1er */                          
             IF         ((&EDTCDE *EQ 'W') *OR (&EDTCDE *EQ 'Y')) DO             
                IF         (&Z *EQ 11) CHGVAR &PRTSW 'Y'                         
             ENDDO                                        /* W ou Y edit code */ 
             IF         ((&PRTSW *EQ 'Y') *OR +                                  
                           (%SST(&NUMVARC &X 1) *NE '0')) DO           /* Prt */ 
              CHGVAR     %SST(&CHARVAR &Y 1) %SST(&NUMVARC &X 1)                 
              CHGVAR     &Y (&Y + 1)                    /* Màj Index sortie */   
              CHGVAR     &PRTSW 'Y'                    /* Switch Print *ON */    
                                                                                 
         /* EDTCDE J et K mise en place du séparateur de milliers */             
               IF         ((&EDTCDE *EQ 'J') *OR (&EDTCDE *EQ 'K')) DO           
                IF         ((&Z *EQ 3) *OR (&Z *EQ 6) *OR (&Z *EQ 9) +           
                           *OR (&Z *EQ 12)) DO                  /* Empl. sép. */ 
                 CHGVAR     %SST(&CHARVAR &Y 1) &SEPCHAR         /* Car. Sép. */ 
                 IF         (&SEPCHAR *NE ' ') CHGVAR &Y (&Y + 1)                
                ENDDO                                                            
               ENDDO                                                             
                                                                                 
         /* Insertion du séparateur Date ou Heure si nécessaire */               
               IF         ((&EDTCDE *EQ 'W') *OR (&EDTCDE *EQ 'Y')) DO           
                 IF         ((&Z *EQ 11) *OR (&Z *EQ 13)) DO      /* Date sép */ 
                   IF         (&EDTCDE *EQ W) CHGVAR %SST(&CHARVAR &Y 1) ':'     
                   IF         (&EDTCDE *EQ Y) CHGVAR %SST(&CHARVAR &Y 1) '/'     
                   CHGVAR     &Y (&Y + 1)            /* Prochaine pos. sortie */ 
                 ENDDO                                                           
               ENDDO                                    /* Sép. date ou heure */ 
             ENDDO                                      /* Print */              
                                                                                 
             CHGVAR     &Z (&Z + 1)                  /* Màj Index sép. symbole*/ 
             CHGVAR     &X (&X + 1)                  /* Index entrée */          
                                                                                 
         /* Force un zéro dans les positions unités */                           
             IF         (&X *EQ 15) CHGVAR &PRTSW 'Y'       /* Force dernier */  
             IF         (&X *NE 16) GOTO LOOP               /* Prochain carac.*/ 
                                                                                 
         /* Si variable négative, insertion du signe */                          
             IF         (&NUMVAR *LT 0) CHGVAR %SST(&CHARVAR &Y 1) '-'           
                                                                                 
 ENDPGM:     ENDPGM                                                              
 


